

Design Study of a Ring Stiffened Cylinder for use as a Manned Submersible

Shell Buckling using Von Mises Equation - Widenberg, D.F and Trilling, C.
 "Collapse by Instability of a Thin Cylindrical Shells Under External Pressure"
 ASME Trans, Volume 56, 1934, P.820, Equation [6]

SafetyFactor := 2.0

DesignGoal := 1320·ft·SafetyFactor

DesignGoal = 2640ft

Design Variables:

Outside Diameter

OD := 42.0·in

SeaWaterDensity := 64· $\frac{\text{lbf}}{\text{ft}^3}$

Shell Thickness

t := .375·in, .4375·in.. .625·in

Shell Length

Len := 104.25·in

Number of Rings

num := 2

Constants:

Material Properties:

Poissons Ratio

$\mu := .3$

Yield Strength

$\sigma := 38000 \cdot \frac{\text{lbf}}{\text{in}^2}$

Youngs Modulus

$E := 30 \cdot 10^6 \cdot \frac{\text{lbf}}{\text{in}^2}$

Equations:

n := 2,3.. 10

$$L := \frac{\frac{1}{3} \cdot \frac{OD}{2} + Len + \frac{1}{3} \cdot \frac{OD}{2}}{num + 1}$$

Mean Diameter

D(t) := OD - t

$$\rho(t,n) := \frac{1}{n^2 \left(\frac{2 \cdot L}{\pi \cdot D(t)} \right)^2 + 1}$$

$$\lambda_2(t,n) := \rho(t,n) \cdot \left[3 + \mu + (1 - \mu^2) \cdot \rho(t,n) \right]$$


$$\lambda_3(t,n) := \rho(t,n) \cdot (1 + \mu) - \left[\rho(t,n)^2 \left[\mu \cdot (1 + 2 \cdot \mu) + (1 - \mu^2) \cdot (1 - \rho(t,n) \cdot \mu) \cdot \left(1 + \frac{1 + \mu}{1 - \mu} \cdot \rho(t,n) \right) \right] \right]$$

$$\mu_1(t,n) := 2 + \lambda_2(t,n)$$

$$\mu_2(t,n) := 1 + \lambda_3(t,n)$$

$$\text{FirstPart}(t,n) := \frac{1}{3} \left[n^2 + \left(\frac{\pi \cdot D(t)}{2 \cdot L} \right)^2 \right]^2 - \mu_1(t,n) \cdot n^2 + \mu_2(t,n) \cdot \frac{2 \cdot E}{(1 - \mu^2)} \cdot \left(\frac{t}{D(t)} \right)^3 + \frac{2 \cdot E \cdot \left(\frac{t}{D(t)} \right)}{\left[n^2 \left(\frac{2 \cdot L}{\pi \cdot D(t)} \right)^2 + 1 \right]^2}$$

$$\text{Depth}(t,n) := \text{FirstPart}(t,n) \cdot \frac{1}{n^2 - 1 + \frac{1}{2} \left(\frac{\pi \cdot D(t)}{2 \cdot L} \right)^2} \cdot \frac{1}{\text{SeaWaterDensity}}$$



$$\frac{\text{Depth}(t, 5)}{\text{ft}} = \begin{pmatrix} 1487 \\ 2167 \\ 3051 \\ 4169 \\ 5553 \end{pmatrix} \quad \frac{t}{\text{in}} = \begin{pmatrix} 0.375 \\ 0.4375 \\ 0.5 \\ 0.5625 \\ 0.625 \end{pmatrix}$$

OD := 39.in, 40.in.. 45.in

t := .5.in

n := 2, 3.. 10

$$L(OD) := \frac{\frac{1}{3} \cdot \frac{OD}{2} + Len + \frac{1}{3} \cdot \frac{OD}{2}}{num + 1} * \quad \text{Mean Diameter} \quad D(OD) := OD - t$$

$$\rho(OD, n) := \frac{1}{n^2 \left(\frac{2 \cdot L(OD)}{\pi \cdot D(OD)} \right)^2 + 1} *$$

$$\lambda 2(OD, n) := \rho(OD, n) \cdot [3 + \mu + (1 - \mu^2) \cdot \rho(OD, n)] *$$

$$\lambda 3(OD, n) := \rho(OD, n) \cdot (1 + \mu) - \left[\rho(OD, n)^2 \cdot \left[\mu \cdot (1 + 2 \cdot \mu) + (1 - \mu^2) \cdot (1 - \rho(OD, n) \cdot \mu) \cdot \left(1 + \frac{1 + \mu}{1 - \mu} \cdot \rho(OD, n) \right) \right] \right] *$$


$$\mu 1(OD, n) := 2 + \lambda 2(OD, n) *$$

$$\mu 2(OD, n) := 1 + \lambda 3(OD, n) *$$

$$\text{FirstPart}(OD, n) := \frac{1}{3} \cdot \left[\left[n^2 + \left(\frac{\pi \cdot D(OD)}{2 \cdot L(OD)} \right)^2 \right]^2 - \mu 1(OD, n) \cdot n^2 + \mu 2(OD, n) \right] \cdot \frac{2 \cdot E}{(1 - \mu^2)} \cdot \left(\frac{t}{D(OD)} \right)^3 *$$

$$\text{SecondPart}(OD, n) := \text{FirstPart}(OD, n) + \frac{2 \cdot E \cdot \left(\frac{t}{D(OD)} \right)}{\left[n^2 \cdot \left(\frac{2 \cdot L(OD)}{\pi \cdot D(OD)} \right)^2 + 1 \right]^2} *$$

$$\text{Depth}(OD, n) := \text{SecondPart}(OD, n) \cdot \frac{1}{n^2 - 1 + \frac{1}{2} \cdot \left(\frac{\pi \cdot D(OD)}{2 \cdot L(OD)} \right)^2} \cdot \frac{1}{\text{SeaWaterDensity}} *$$



$$\frac{OD}{in} =$$

39
40
41
42
43
44
45

$$\frac{\text{Depth}(OD, 5)}{\text{ft}} = \begin{pmatrix} 3516 \\ 3342 \\ 3188 \\ 3051 \\ 2930 \\ 2822 \\ 2727 \end{pmatrix}$$